

Typed Tagless Final Interpreters

Stuart Terrett

May 11, 2020

preamble

```
{-# LANGUAGE GADTs #-}  
module TTFI where
```

typed

The object language is *typed*

typed

The object language is *typed*

```
add (int 2) (bool 3)
```

typed

The object language is *typed*

```
add (int 2) (bool 3)
```

will not compile

The object language terms are not wrapped in a sum type with type tags

tagless

The object language terms are not wrapped in a sum type with type tags

```
data Term =  
    TInt Int  
    | TBool Bool
```

tagless

The object language terms are not wrapped in a sum type with type tags

```
data Term =  
  TInt Int  
  | TBool Bool
```

The object language is "tight": if it compiles, it cannot get stuck

```
addU :: Term → Term → Term  
addU (TInt x) (TInt y) = TInt $ x + y  
addU _ _ = error "make illegal states unrepresentable"
```


final

Terms of the object language are represented as expressions in the metalanguage

final

Terms of the object language are represented as expressions in the metalanguage

$$\begin{aligned} twoPlusTwo &:: (AddSym\ repr) \Rightarrow repr\ Int \\ twoPlusTwo &= add\ (int\ 2)\ (int\ 2) \end{aligned}$$

final

Terms of the object language are represented as expressions in the metalanguage

$$\begin{aligned} twoPlusTwo &:: (AddSym\ repr) \Rightarrow repr\ Int \\ twoPlusTwo &= add\ (int\ 2)\ (int\ 2) \end{aligned}$$

Not abstract syntax

```
data AST =  
  Lit Term  
  | AddU AST AST  
  | AndU AST AST  
  
twoPlusTwo' :: AST  
twoPlusTwo' = AddU (Lit (TInt 2)) (Lit (TInt 2))
```

interpreter

The syntax is a typeclass

```
class AddSym repr where  
  int :: Int → repr Int  
  add :: repr Int → repr Int → repr Int
```

typed tagless

The tagless-typed-ness comes from

$$\textit{repr} :: * \rightarrow *$$

typed tagless

The tagless-typed-ness comes from

$$\text{repr} :: * \rightarrow *$$

Which is why the above example won't typecheck

$$\text{int } 2 :: \text{repr } \text{Int}$$
$$\text{bool } 3 :: \text{repr } \text{Bool}$$
$$\text{add} :: \text{repr } \text{Int} \rightarrow \text{repr } \text{Int} \rightarrow \text{repr } \text{Int}$$

interpreter

The semantics are an instance of the typeclass

```
newtype  $R\ a = R\ \{unR :: a\}$   
instance  $AddSym\ R$  where  
   $int = R$   
   $add\ (R\ x)\ (R\ y) = R\ \$\ x + y$ 
```

Adding interpretations

```
newtype S a = S { unS :: Int → String }  
instance AddSym S where  
  int = S ∘ const ∘ show  
  add (S x) (S y) = S $ λc →  
    "(" <> x c <> " + " <> y c <> ")"
```


extensible

Adding operations

```
class MultiSym repr where  
  mul :: repr Int → repr Int → repr Int  
  
instance MultiSym R where  
  mul (R x) (R y) = R $ x * y  
  
instance MultiSym S where  
  mul (S x) (S y) = S $ λc →  
    "(" <> x c <> " * " <> y c <> ")"
```

adding booleans

```
class BoolSym repr where  
  bool :: Bool → repr Bool  
  lte :: repr Int → repr Int → repr Bool  
  when :: repr Bool → repr a → repr a → repr a
```

adding booleans

instance *BoolSym* *R* **where**

bool = *R*

lte (*R* *x*) (*R* *y*) = *R* \$ *x* ≤ *y*

when (*R* *b*) (*R* *t*) (*R* *f*) = *R* \$ **if** *b* **then** *t* **else** *f*

adding booleans

instance *BoolSym* *R* **where**

bool = *R*

lte (*R* *x*) (*R* *y*) = *R* \$ *x* ≤ *y*

when (*R* *b*) (*R* *t*) (*R* *f*) = *R* \$ **if** *b* **then** *t* **else** *f*

instance *BoolSym* *S* **where**

bool = *S* ∘ *const* ∘ *show*

lte (*S* *x*) (*S* *y*) = *S* \$ λ*c* →

"(" <> *x* *c* <> " <= " <> *y* *c* <> ")"

when (*S* *b*) (*S* *t*) (*S* *f*) = *S* \$ λ*c* →

"if " <> *b* *c*

<> "\nthen " <> *t* *c*

<> "\n else" <> *f* *c*

adding lambda abstraction

```
class LamSym repr where  
  lam :: (repr a → repr b) → repr (a → b)  
  app :: repr (a → b) → repr a → repr b
```

adding lambda abstraction

instance *LamSym* *R* **where**
 lam *f* = *R* \$ *unR* ∘ *f* ∘ *R*
 app (*R* *f*) (*R* *a*) = *R* \$ *f* *a*

adding lambda abstraction

instance *LamSym* *R* **where**

lam *f* = *R* \$ *unR* \circ *f* \circ *R*

app (*R* *f*) (*R* *a*) = *R* \$ *f* *a*

instance *LamSym* *S* **where**

lam *f* = *S* \$ $\lambda c \rightarrow$

let

x = "x" <> *show* *c*

in

"\\" <> *x* <> ".\n"

<> (*unS* \circ *f* \circ *S* \$ *const* *x*) (*c* + 1)

app (*S* *f*) (*S* *a*) = *S* \$ $\lambda c \rightarrow$

"(" <> *f* *c* <> " " <> *a* *c* <> ")"

higher-order abstract syntax

Notice that the metalanguage (Haskell) maintains the variable bindings

adding recursion!!

```
class FixSym repr where  
  fix :: (repr a → repr a) → repr a
```

adding recursion!!

```
instance FixSym R where  
  fix f = R $ fx (unR ∘ f ∘ R)  
  where fx g = g (fx g)
```

adding recursion!!

```
instance FixSym R where  
  fix f = R $ fx (unR ∘ f ∘ R)  
  where fx g = g (fx g)
```

```
instance FixSym S where  
  fix f = S $ λc →  
    let  
      self = "self" <> show c  
    in  
      "(fix " <> self <> ".\n"  
      <> (unS ∘ f ∘ S $ const self) (c + 1) <> ")"
```

simply typed lambda calculus with integer and boolean literals

```
class (AddSym repr  
    , MultSym repr  
    , BoolSym repr  
    , LamSym repr  
    , FixSym repr  
    )  $\Rightarrow$  Symantics repr
```

```
instance Symantics R
```

```
instance Symantics S
```

```
eval :: R a  $\rightarrow$  a
```

```
eval = unR
```

```
pprint :: S a  $\rightarrow$  String
```

```
pprint e = unS e 0
```

factorial

```
factorial :: (Symantics repr) => repr (Int → Int)
factorial = fix (λself → lam (λn →
    when (lte n (int 0))
        (int 1)
        (mul n (self 'app' (add n (int (-1)))))))
```

(hint: now's a good time to `stack ghci src/TTFI.hs`)

contrast: typed tagless initial encoding

data *Syml* *h* *a* **where**

INT :: *Int* → *Syml* *h* *Int*

Add :: *Syml* *h* *Int* → *Syml* *h* *Int* → *Syml* *h* *Int*

Mul :: *Syml* *h* *Int* → *Syml* *h* *Int* → *Syml* *h* *Int*

BOOL :: *Bool* → *Syml* *h* *Bool*

Lte :: *Syml* *h* *Int* → *Syml* *h* *Int* → *Syml* *h* *Bool*

When :: *Syml* *h* *Bool* → *Syml* *h* *a* → *Syml* *h* *a* → *Syml* *h* *a*

Var :: *h* *a* → *Syml* *h* *a*

Lam :: (*Syml* *h* *a* → *Syml* *h* *b*) → *Syml* *h* (*a* → *b*)

App :: *Syml* *h* (*a* → *b*) → *Syml* *h* *a* → *Syml* *h* *b*

Fix :: (*Syml* *h* *a* → *Syml* *h* *a*) → *Syml* *h* *a*

h is the evaluation context (eg *R* or *S* etc)

Like the final encoding, this is well typed (if it compiles, it won't get stuck) and there are no "type tags".

Like the final encoding, this is well typed (if it compiles, it won't get stuck) and there are no "type tags".

This requires **GADTs** so that the result of pattern-matching on different branches of a sum type can have different types

initial interpreter

Here we evaluate in **R**

$eval :: \text{Sym} \rightarrow R \rightarrow a \rightarrow a$

$eval (INT\ i) = i$

$eval (Add\ x\ y) = (eval\ x) + (eval\ y)$

$eval (Mul\ x\ y) = (eval\ x) * (eval\ y)$

$eval (BOOL\ b) = b$

$eval (Lte\ x\ y) = (eval\ x) \leq (eval\ y)$

$eval (When\ b\ t\ f) = \text{if } (eval\ b) \text{ then } (eval\ t) \text{ else } (eval\ f)$

$eval (Var\ x) = unR\ x$

$eval (Lam\ f) = eval \circ f \circ Var \circ R$

$eval (App\ f\ a) = (eval\ f) (eval\ a)$

$eval (Fix\ f) = eval\ (fx\ f)$

where $fx\ g = g\ (fx\ g)$

expression problem

Adding interpreters to `SymI` is straightforward

```
pprintI :: Sym S a → String  
pprintI = ...
```

expression problem

Adding interpreters to `SymI` is straightforward

```
pprintI :: Sym S a → String  
pprintI = ...
```

But adding new operations isn't possible

expression problem

Adding interpreters to `SymI` is straightforward

```
pprintI :: Sym S a → String  
pprintI = ...
```

But adding new operations isn't possible

Because of the contravariant `SymI` in `Lam`, `SymI` is not representable as the fixpoint of a functor, and so we can't use the coproduct-of-functors representation.

bijection

We can show the final and initial embeddings are equivalent by establishing a bijection

final to initial

instance *AddSym* (*SymI* *h*) **where**

int = *INT*

add = *Add*

instance *MultSym* (*SymI* *h*) **where**

mul = *Mul*

instance *BoolSym* (*SymI* *h*) **where**

bool = *BOOL*

lte = *Lte*

when = *When*

instance *LamSym* (*SymI* *h*) **where**

lam = *Lam*

app = *App*

instance *FixSym* (*SymI* *h*) **where**

fix = *Fix*

instance *Symantics* (*SymI* *h*)

fTol :: *SymI* *h* *a* → *SymI* *h* *a*

fTol = *id*

initial to final

$iToF :: (Symantics\ repr) \Rightarrow Sym1\ repr\ a \rightarrow repr\ a$

$iToF\ (INT\ x) = int\ x$

$iToF\ (Add\ x\ y) = add\ (iToF\ x)\ (iToF\ y)$

$iToF\ (Mul\ x\ y) = mul\ (iToF\ x)\ (iToF\ y)$

$iToF\ (BOOL\ b) = bool\ b$

$iToF\ (Lte\ x\ y) = lte\ (iToF\ x)\ (iToF\ y)$

$iToF\ (When\ b\ t\ f) = when\ (iToF\ b)\ (iToF\ t)\ (iToF\ f)$

$iToF\ (Var\ a) = a$

$iToF\ (Lam\ f) = lam\ (\lambda x \rightarrow iToF\ (f\ (Var\ x)))$

$iToF\ (App\ f\ a) = app\ (iToF\ f)\ (iToF\ a)$

$iToF\ (Fix\ f) = fix\ (\lambda self \rightarrow iToF\ (f\ (Var\ self)))$

end

Check out Oleg's writeups at

- ▶ <http://okmij.org/ftp/tagless-final/index.html>
- ▶ <http://okmij.org/ftp/tagless-final/course/index.html>

This presentation and some code snippets at
<https://github.com/shterrett/ttfi>